Discontinuity and Angles

In this note, we will use Mathematica to

- Visualize discontinuous behavior of complex functions along a curve.
- Visualize the angle between two curves.

1 Discontinuous functions

Let us consider the function $f(z) = \mathrm{Log}(z^2+1)$. From the Mathematica practice last time, we know that f is continuous everywhere excepts for points on the line $x = 0, y \geq 1$ and the line $x = 0, y \leq -1$. *We want to see how $f(z)$ jumps as z crosses the imaginary axis* (Figure 1). Let us



Figure 1

consider the line $y = 2$, which has complex form $z = t + 2i$. This line to mapped to some curve by f, which we call an *image curve*. We expect that when t moves from -2 to 2, the image curve is drawn out continuously, except at $t = 0$. At $t = 0$, the function $f(z) = \ln|z^2+1| + i\mathrm{Arg}(z^2+1)$ jumps by $2\pi i$. This is a jump of distance 2π on the vertical direction. In Mathematica (Figure 2),

```
f[z_] := Log[z^2 + 1]
p[s_] := ParametricPlot[ReIm[t+2*I],{t,-2,s},PlotRange -> {{-2,2},{0,4}}]
q[s_] := ParametricPlot[ReIm[f[t+2*I]],{t,-2,s},PlotRange -> {{0,2.5},{-3,3}}]
Manipulate[{p[s],q[s]},{s,-1.9,2}]
```



Figure 2

Note that there is no mystery about the numbers -2, 2, 0, 4, 2.5, -3, 3 which we put in **PlotRange**. One should omit option **PlotRange** from the above commands at the first time of running. This will cause the frame of the plot to vary as one varies s. Once we know the maximum range of the plot, we can specify the **PlotRange** to fix the frame of the plot.

2 Visualize angles between two curves

Put $z_0 = 1 + i$. There are infinitely many curves on the complex plane that pass through z_0. To make vivid our experiment, let us consider two families of curves that pass through z_0.

$$\begin{aligned}\sigma_s(t) &= z_0 + e^{is}(1 - e^{it}) \\ \lambda_s(t) &= z_0 + t(s + i\cos t).\end{aligned}$$

For each value of s, the curve σ_s and λ_s pass through z_0 when $t = 0$. One can plot both curves together on the complex plane as follows (Figure 3).

```
z0 = 1+I
sigma[s_,t_] := z0 + Exp[I*s]*(1-Exp[I*t])
lambda[s_,t_] := z0 + t*(s+I*Cos[t])
p[s_] := ParametricPlot[{ReIm[sigma[s,t]], ReIm[lambda[s,t]]},
    {t,-2,2}, PlotRange -> {{0,2.5},{0,2}}, PlotLegends -> Automatic]
Manipulate[p[s],{s,-1,1}]
```



Figure 3

Note that the option **PlotLegends** is for us to distinguish the curves more easily. The blue curve corresponds to the first function (which is σ_s) and the orange curve corresponds to the second function (which is λ_s).

Now let us draw the velocity vectors on σ_s at λ_s at z_0. That is to draw vectors $\sigma_s'(0)$ and $\lambda_s'(0)$ from the point z_0. The option **Epilog** $\rightarrow$ **{. . . }** allows us to annotate the graph. We want to draw two arrows, namely $\sigma_s'(0)$ and $\lambda_s'(0)$, at point z_0. In Mathematica, the command **Arrow**[**{{a,b},{c,d}}**] draws an arrow from point (a, b) to point (c, d). One can compute

$$\begin{aligned}\sigma_s'(0) &= e^{is}(-i) \\ \lambda_s'(0) &= s + i.\end{aligned}$$

Thus, the first arrow $\sigma_s'(0)$ can be drawn by the command

Arrow[{{1,1}, {1,1} + ReIm[Exp[I*s]*(-I)]}]

The second arrow can be drawn by the command

Arrow[{{1,1}, {1,1} + ReIm[s+I]}]

Don't execute those commands yet. We put these two commands inside the curly brackets (separated by comma) of the **Epilog→{. . . }** command as follows (Figure 4).

```
p[s_] := ParametricPlot[{ReIm[sigma[s,t]], ReIm[lambda[s,t]]},
    {t,-2,2}, PlotRange -> {{0,2.5},{0,2}}, PlotLegends -> Automatic,
    Epilog -> {Arrow[{{1,1}, {1,1} + ReIm[Exp[I*s]*(-I)]}],
    Arrow[{{1,1}, {1,1} + ReIm[s+I]}]}]
Manipulate[p[s],{s,-1,1}]
```



Figure 4

The angle between the two curves at the intersection point z_0 is defined as the angle between these two velocity vectors. To compute the angle between two curves σ_s and λ_s, for example when $s = -1$ (see Figure 5), we first compute the argument of each velocity vector:

$$\begin{aligned}\theta_1 &= \text{Arg}(\sigma_s'(0)) = \text{Arg}(e^{is}(-i)) = \text{Arg}(e^{-i}(-i)) \\ \theta_2 &= \text{Arg}(\lambda_s'(0)) = \text{Arg}(s+i) = \text{Arg}(-1+i).\end{aligned}$$

Then the angle between the two velocity vectors (sweeping from $\sigma_s'(0)$ to $\lambda_s'(0)$) is $\theta = \theta_2 - \theta_1$ (in modulo 2π). In Mathematica,

```
theta1 = Arg[Exp[-I]*(-I)]
theta2 = Arg[-1+I]
theta = theta2 - theta1
```



Figure 5

Consider the function

$$f(z) = \frac{iz}{z-3}.$$

We want to see the angle between the image of the curve σ_s and the image of the curve λ_s under f. The image of the curve σ_s under f is $\Sigma_s(t) = f(\sigma_s(t))$. The image of the curve λ_s under f is $\Lambda_s(t) = f(\lambda_s(t))$. The image of z_0 under f is $w_0 = f(z_0)$. In Mathematica,

```
f[z_] := I*z/(z-3)
w0 = f[z0]
```

The velocity of Σ_s at w_0 is $\Sigma_s'(0)$. The velocity of Λ_s at w_0 is $\Lambda_s'(0)$. In Mathematica, one can compute these velocity vectors by

```
Sigma[s_,t_] := f[sigma[s,t]]
Lambda[s_,t_] := f[lambda[s,t]]
v1[s_] := D[Sigma[s,t],t] /. t -> 0
v2[s_] := D[Lambda[s,t],t] /. t -> 0
```

Here the operator `/.` is the substitution operator. The third of the above commands means that the velocity vector $v_1(s)$ is obtained by first taking the derivative of $\Sigma(s,t)$ with respect to t and then substituting t by 0.

Because f is holomorphic at z_0 and

$$f'(z_0) = \left.\frac{-3i}{(z-3)^2}\right|_{z=z_0} = \frac{-3i}{(i-2)^2} \neq 0,$$

we know that f is angle-preserving (conformal) at z_0. In Mathematica,

```
p[s_] := ParametricPlot[{ReIm[sigma[s,t]], ReIm[lambda[s,t]]},
    {t,-2,2}, PlotRange -> {{0,2.5},{0,2}},
    Epilog -> {Arrow[{ReIm[z0], ReIm[z0] + ReIm[Exp[I*s]*(-I)]}],
    Arrow[{ReIm[z0], ReIm[z0] + ReIm[s+I]}]}]

q[s_] := ParametricPlot[{ReIm[Sigma[s,t]], ReIm[Lambda[s,t]]},
    {t,-2,2}, PlotRange -> {{0,3},{-2,2}},
    Epilog -> {Arrow[{ReIm[w0], ReIm[w0] + ReIm[v1[s]]}],
    Arrow[{ReIm[w0], ReIm[w0] + ReIm[v2[s]]}]}]

Manipulate[{p[s],q[s]},{s,-1,1}]
```



Figure 6

We can see from the picture that the angle between the image curves Σ_s and Λ_s is the same as the angle between the original curves σ_s and λ_s.